



Replacing Nested IF Functions with a Lookup

BY BILL JELEN

In Excel 2007, Microsoft increased the limit of nested IF functions from seven to 32. While I guess we can applaud any improvement that Microsoft offers, this is one feature improvement that I hope you will never use. Allowing 32 nested IF functions enables people to

build formulas that are impossibly difficult to understand. The truth is that an obscure version of the VLOOKUP formula would solve the problem long before someone hits the old limit of seven nested IF functions.

Let's say that the VP of Sales proposes a sales promotion for the month of February. The sales reps will be paid a 3% bonus any time they bring in a sale where the gross profit percent (GP%) is greater than 55%. The IF function is perfect for calculating this bonus. The syntax of the function is:

```
=IF(Logical Test, Value if True, Value if False)
```

Using the worksheet in Figure 1, the bonus calculation would be:

```
=IF(F2>=0.55,0.03*C2,0)
```

The formula says, "If the GP% in

F2 is greater than 55%, then calculate the bonus as 3% of the revenue in C2; otherwise, the bonus is 0." At the end of February, you calculate the bonuses using the IF function, and it turns out the program worked: Fewer reps offered discounts, overall gross profit improved by a few points, and everyone was happy.

The VP of Sales then decides he can improve the bonus program. In April, he raises the bar, offering a 3% bonus when the GP% is over 58% and a 1% bonus when the GP% is over 55%. A natural reaction is to embed a new IF function in place of the zero in the original function:

```
=IF(F2>=0.58,0.03,IF(F2>=0.55,0.01,0))*C2
```

The first IF function says if the GP% is greater than 58%, then use a 3% bonus. For the "Value if False"

component of the first IF function, a second IF function says that the bonus is 1% if the GP% is greater than 55%. While this function is longer and a bit more complicated, you can still easily trace what is happening.

As the months go by, the VP keeps adding more and more bonus levels. By October, the formula to calculate the bonus offers seven IF functions:

```
=IF(F2>=0.63,0.05,IF(F2>=0.61,0.045,IF(F2>=0.6,0.04,IF(F2>=0.59,0.03,IF(F2>=0.58,0.025,IF(F2>=0.57,0.02,IF(F2>=0.55,0.01,0))))))*C2
```

This formula is at the limit of seven nested IF functions. If any new bonus levels are offered in December, Excel 2003 won't accept additional nested IF functions. Excel 2007, however, will. While this would enable the Sales department to come up with increasingly complex bonus plans, I don't recommend building an IF function with 32 nested IFs. By the time that most people hit the old limit of seven, it would make more sense to take advantage of a little-known feature of the VLOOKUP function.



Figure 1

	A	B	C	D	E	F	G	H	I	J	K	L	M	N
1	Invoice	Rep	Revenue	Cost	Profit	GP%	Bonus	Rep Name		GP%	Bonus		Rep	Name
2	43666	R14	187,094	77,831	109,263	58.4%	4,677	Barber		0.00	0.000		R29	Alexander
3	43667	R32	62,646	39,718	22,928	36.6%	0	Payne		0.55	0.010		R14	Barber
4	43668	R03	149,716	58,240	91,476	61.1%	6,737	Patton		0.57	0.020		R27	Barnes
5	43669	R34	156,445	76,032	80,413	51.4%	0	Marsh		0.58	0.025		R04	Beach
6	43670	R38	125,304	64,532	60,772	48.5%	0	Medina		0.59	0.030		R44	Bentley
7	43671	R26	37,739	17,284	20,455	54.2%	0	Brady		0.60	0.040		R26	Brady
8	43672	R19	140,063	52,664	87,399	62.4%	6,303	Serrano		0.61	0.045		R01	Dominguez
9	43673	R21	78,803	32,940	45,863	58.2%	1,970	Salinas		0.63	0.050		R08	Houston
10	43674	R36	40,672	16,594	24,078	59.2%	1,220	Martinez					R23	Livingston

The Traditional Use of VLOOKUP

Ever since you switched from Lotus to Excel more than a decade ago, you've been writing your VLOOKUP functions with a "False" at the end. While I always imagined this fourth argument was designed to drive former Lotus users crazy, it turns out it empowers an interesting alternative to the typical use of VLOOKUP. In accounting, you're usually looking for the exact match of a certain value. For example, to find the Rep Name for a particular Rep Number, you might paste a table of Rep Names and Numbers to the right of your data set. To copy the correct Rep Name to each record, you would use a formula such as:

```
=VLOOKUP(B2,$M$2:$N$24,2,
FALSE)
```

In the traditional use of VLOOKUP, the yellow table in columns M and N doesn't need to be sorted. If Excel doesn't find an exact match for the

Rep Number in column M, the function will return #N/A.

The Range Version of VLOOKUP

If you omit the fourth argument from the VLOOKUP, it has a different functionality. Instead of finding an exact match in the first column of the lookup table, Excel now looks for the closest value that is less than or equal to the lookup value.

When you use numeric ranges in the lookup table, the table must be sorted in ascending order based on the key field. This forces you to rearrange the sales department's logic a bit. In Figure 1, cells J2:K2 start the table by saying that if the GP% is 0%, then the bonus percentage is 0%.

Cells J3:K3 say that the bonus percentage is 1% once the GP% reaches 55%. Cells J4:K4 indicate that once the GP% reaches 57%, then the bonus percentage is 2%. Continue converting the old formula by changing each nested IF function to

a new row in the lookup table. The seven nested IF functions in the October formula convert to the table shown in cells J2:K9 of Figure 1. As the Sales department continues to add new bonus levels, you can simply add new rows to the table in column J. The simplified formula to calculate the bonus is:

```
=VLOOKUP(F2,$J$2:$K$9,2)*C2
```

I can still remember my original annoyance with Excel that the VLOOKUP function required the extra "FALSE" at the end in order to work like the similar function in Lotus. It turns out that Excel's version of VLOOKUP offers some interesting tricks that can dramatically simplify your nested IF functions. ■

Bill Jelen will appear at a Saturday pre-Conference workshop at the IMA 89th Annual Conference & Exposition. Send questions for future articles to IMA@MrExcel.com.